
COMP2121: Microprocessors and 
Interfacing

Interrupts 

http://www.cse.unsw.edu.au/~cs2121

Lecturer:  Hui Wu

Term 2, 2019

Overview
• Interrupt System Specifications

• Multiple Sources of Interrupts

• Interrupt Priorities

• Polling

• AVR Interrupts 

• Interrupt Vector Table

• System Reset

• Watchdog Timer

• Timer/Counter0

• External Interrupts

• Interrupt Service Routines

Processor

Computer

Memory
(where 
programs, 
data live 
when
running)

Input
Devices

Output
Devices

Keyboard, 
Mouse

Display, 
Printer

Disk
(where 
programs, 
data live 
when not 
running)

Major Components of any Computer

How CPU Interacts with I/O?

Two Choices:

• Interrupts.
 I/O devices generate signals to request services from CPU .

 Need special hardware to implement interrupts.

 Efficient.
 A signal is generated only if the I/O device needs services from CPU. 

• Polling
 Software queries I/O devices.

 No hardware needed.

 Not efficient.  
 CPU may waste processor cycles to query a device even if it does not 

need any service.

Interrupt System Specifications (1/2)  

1. Allow for asynchronous events to occur and be recognized.

2. Wait for the current instruction to finish before taking care 
of any interrupt.

3. Branch to the correct interrupt handler, also called interrupt 
service routine, to service the interrupting device.

4. Return to the interrupted program at the point it was 
interrupted.

5. Allow for a variety of interrupting signals, including levels 
and edges.

6. Signal the interrupting device with an acknowledge signal 
when the interrupt has been recognized.  

Interrupt System Specifications (2/2) 

7. Allow programmers to selectively enable and disable all 
interrupts.

8. Allow programmers to enable and disable selected 
interrupts.

9. Disable further interrupts while the first is being serviced

10. Deal with multiple sources of interrupts.

11. Deal with multiple, simultaneous interrupts.  

Interrupt Recognition and Ack (1/3)

Interrupt signal to 
sequence 
controller

Interrupt ack from 
sequence 
controller

SEQUENCE 
CONTROLLER

Disable interrupt 
instruction

Enable interrupt 
instruction

Return from 
interrupt 
instruction

INTERR-
UPTING 
DEVICE

Signal 
conditioning

Set
IRQ-FF
Reset

Set
INTE-FF
Reset

IRQ

Interrupt 
Enable

CPU

Pending 
Interrupt

Interrupt Recognition and Ack (2/4)

• An Interrupt Request (IRQ) may occur at any time.

 It may have rising or falling edges or high or low levels.

 Frequently it is a active-low signal and multiple devices are wire-
ORed together.

• Signal Conditioning Circuit detects these different types of 
signals.

• Interrupt Request Flip-Flop (IRQ-FF) remembers that an 
interrupt request has been generated until it is acknowledged.

When  IRQ-FF is set, it generates a pending interrupt signal that goes 
towards the Sequence Controller.

 IRQ-FF is reset when CPU acknowledges the interrupt with INTA 
signal.

Interrupt Recognition and Ack (3/4)

• The programmer has control over interrupting process by 
enabling and disabling interrupts with explicit instructions

 The hardware that allows this is Interrupt Enable Flip-Flop (INTE-
FF).

When the INTE-FF is set, all interrupts are enabled and the pending 
interrupt interrupt is allowed through the AND gate to the sequence 
controller.

 The INTE-FF is reset in the following cases. 

 CPU acknowledges the interrupt.

 CPU is reset.

 Disable interrupt instruction is executed. 

Interrupt Recognition and Ack (4/4)

• An interrupt acknowledge signal is generated by the CPU 
when the current instruction has finished execution and CPU has 
detected the IRQ. 

 This resets the IRQ-FF and INTE-FF and signals the interrupting 
device that CPU is ready to execute the interrupting device routine.

• At the end of the interrupt service routine, CPU executes a 
return-from-interrupt instruction.

 Part of this instruction’s job is to set the INTE-FF to reenable 
interrupts.

 If the IRQ-FF is set during an interrupt service routine a pending 
interrupt will be recognized by the sequence controller immediately after 
the INTE-FF is set.  This allows nested interrupts i.e. interrupts 
interrupting interrupts. 

Multiple Sources of Interrupts

• Determine which of the multiple devices has generated the IRQ to be 
able to execute its interrupt service routine.

 Two approaches: Polled interrupts and vectored interrupts.

• Resolve simultaneous requests from interrupts with a prioritization 
scheme.

CPU Device 1 Device 2 Device n• • •

INTA
IRQ

Data

Address

Control 

Polled Interrupts

• Software, instead of hardware,  is responsible for 
determining the interrupting device.

 The device must have logic to generate the IRQ signal and to set 
an “I did it” bit  in a status register that is read by CPU. 

 The bit is reset after the register has been read.

• IRQ signals the sequence controller to start executing an 
interrupt service routine that first polls the device then 
branches to the correct service routine.

Polled Interrupt Logic

Logic to 
generate IRQ

Logic to reset IRQ 
when status register 
is read

Logic to read status 
register and reset “I 
did it” bit

Logic to set “I 
did it” bit

Status register

Data

Address

Control 

IRQ

Vectored Interrupts (I)

• CPU’s response to IRQ is to assert INTA.

• The interrupting device uses INTA to place information 
that identifies itself, called vector,  onto the data bus for 
CPU to read.

 An vector is the address of an interrupt service routine.  

• CPU uses the vector to execute the interrupt service 
routine.

Vectored Interrupting Device 
Hardware (I)

Logic to 
generate IRQ

Logic to reset 
IRQ

Vector Information

Three-State Driver

Data

Address

Control 

INTA
IRQ

Vector Interrupts (II)

• CPU has multiple IRQ input pins.

• CPU designers reserve specific memory locations for a vector 
associated with each IRQ line.

• Individual disable/enable bit is assigned to each interrupting source. 

CPU

•••

IRQ 0

IRQ 1

IRQ 2

IRQ n

Interrupt Priorities

• When multiple interrupts occurs at the same time, 
which one will be serviced first?

• Two resolution approaches:

 Software resolution

 Polling software determines which interrupting source is 
serviced first.

 Hardware resolution

 Daisy chain.

 Separate IRQ lines.

 Hierarchical prioritization.

 Nonmaskable interrupts. 

Daisy Chain Priority Resolution (1/2)

• CPU asserts INTA that is passed down the chain from 
device to device. The higher-priority device is closer to 
CPU.  

• When the INTA reaches the device that generated the 
IRQ,  that device puts its vector on the data bus and not 
passing along the INTA. So lower-priority devices do 
NOT receive the INTA.  

Daisy Chain Priority Resolution (2/2)

CPU Device 1 Device 2 Device n• • •

IRQ

INTA INTA INTA INTA

Data

Address

Control 

Hardware Priority Resolution

• Separate IRQ Lines.

 Each IRQ line is assigned a fixed priority. For example, 
IRQ0 has higher priority than IRQ1 and IRQ1 has higher 
priority than IRQ2 and so on.  

• Hierarchical Prioritization.

 Higher priory interrupts are allowed while lower ones are 
masked.

• Nonmaskable Interrupts.

 Cannot be disabled.

 Used for important events such as power failure.

Transferring Control to 
Interrupt Handler

• Hardware needs to save the return address.

 Most processors save the return on the stack.

 ARM uses a special register, link register, to store the return 
address. 

• Hardware may also save some registers such as program 
status register. 

 AVR does not save any register. It is programmer’s responsibility 
to save program status register and conflicting registers.

• The delay from the time the IRQ is generated by the 
interrupting device to the time the interrupt handler starts 
to execute is called interrupt latency.

Interrupt Handler

• A sequence of code to be executed when the corresponding 
interrupt is responded by CPU.

• Consists of three parts: Prologue, Body and Epilogue.

• Prologue: 

 Code for saving conflicting registers on the stack.

• Body: 

 Code for doing the required task. 

• Epilogue:

 Code for restoring all saved registers from the stack.

 The last instruction is the return-from-interrupt instruction. 

 reti in AVR.

Software Interrupt 

• Software interrupt is the interrupt generated by software 
without a hardware-generated-IRQ.

• Software interrupt is typically used to implement system calls 
in OS. 

• Most processors provide a special machine instruction to 
generate software interrupt.

 SWI in ARM.

• AVR does NOT provide a software interrupt instruction.  

 Programmers can use External Interrupts to implement software 
interrupts.

Exceptions 

• Abnormalities that occur during the normal operation of 
the processor.

 Examples are internal bus error, memory access error and 
attempts to execute illegal instructions.

• Some processors handle exceptions in the same way as 
interrupts.

 AVR does not handle exceptions.

Reset 

• Reset is an interrupt in most processors (including AVR).

• It is a signal asserted on a separate pin.

• Nonmaskable.  

• It does not do other interrupt processes, such as saving 
conflict registers. It initialize the system to some initial 
state.

AVR MCU Architecture

Interrupts in AVR Microcontrollers (1/3) 

• The number of interrupts varies with specific AVR devices.

• Two types of interrupts:  Internal interrupts and external 
interrupts.
 Internal interrupts: Generated by on-chip I/O devices.

 External interrupts: Generated by external I/O devices.

• For most internal interrupts, they don’t have an individual 
enable/disable bit.
 Program cannot enable/disable these interrupts. 

• External interrupts have an individual enable/disable bit. 
 Program can enable/disable these interrupts.

 An external interrupt can be rising edge-triggered, or falling edge-
triggered or low level-triggered).
 Special I/O registers (External Interrupt Control Registers EICRA 

and EICRB in ATmega2560) to specify how each external interrupt 
is triggered. 

Interrupts in AVR (2/3) 

• There is a global interrupt enable/disable bit, the  I-bit,  in 
Program Status Register SREG.
 Setting the I-bit will enable all interrupts except those with individual 

enable/disable bit. Those interrupts are enabled only if both I and their 
own enable/disable bit are set.

 The I-bit is cleared when an interrupt occurs and is set by the 
instruction reti. 

 Programmers can use sei and cli to set and clear the I-bit. 

 If the I-bit is enabled in the interrupt service routine, nested interrupts 
are allowed.    

• SREG is not automatically saved by hardware when entering  
an interrupt service routine.
 An interrupt service routine needs to save it and other conflict 

registers on the stack at the beginning  and restore them at the end. 

Interrupts in AVR (3/3) 

• Reset is handled as a nonmaskable interrupt.

• Each interrupt has a 4-byte interrupt vector, containing an 
instruction to be executed after MCU has accepted the 
interrupt.

• Each interrupt vector has a vector number, an integer from 
1 to n, the maximum number of interrupts. 

• The priority of each interrupt is determined by its vector 
number.
 The lower the vector number, the higher priority.

• All interrupt vectors, called Interrupt Vector Table,   are 
stored in a contiguous section in  flash memory.
 Starts from address 0 by default.

 Can be relocated.

Interrupt Vectors in ATmega2560 (1/4) 

29

30



16

Interrupt Vectors in ATmega2560 (2/4) 

Interrupt Vectors in ATmega2560 (3/4) 

Interrupt Vectors in ATmega2560 (4/4) 

Interrupt Vectors in ATmega2560 (2/3) 

33

34



18

Interrupt Vector Initialization in 
ATmega2560 (1/5) 

• Typically an interrupt vector contains a branch instruction
(jmp) that branches to the first instruction of the interrupt
handler, or reti (return from interrupt), indicating that this
interrupt is not handled.

Interrupt Vector Initialization in 
ATmega2560 (2/5) 

Interrupt Vector Initialization in 
ATmega2560 (3/5) 

Interrupt Vector Initialization in 
ATmega2560 (4/5) 

Interrupt Vector Initialization in 
ATmega2560 (5/5) 

An Example (1/2)  
.include "m2560def.inc"

.cseg

.org 0x0000   ; Reset vector is at address 0x0000

jmp RESET     ; Jump to the start of Reset interrupt handler

.org INT0addr ; INT0addr is the address of INT0 defined in m2560def.inc

jmp IRQ0        ; Jump to the start of the interrupt handler of IRQ0

.org INT1addr ; INT1addr is the address of INT1 defined in m2560def.inc

reti                 ; Return to the breakpoint where INT1 occurred

…

.org 0x0072    ; Next instruction start at 0x0072  

IRQ0: push r0  ; Save r0

push r1  ; Save r1

…          ; Body of the interrupt handler of INT0

pop r1   ; Restore r1

pop r0   ; Restore r0

reti        ; Return to the breakpoint where INT0 occurred

An Example (2/2)  

RESET: ldi r16, high(RAMEND)   ; Interrupt handler for RESET

out SPH, r16                            ; Set the stack pointer SP to the top of data memory

ldi r16, low(RAMEND)

out SPL,r16

sei                                          ; Enable interrupts

main:                                      ; Main program starts here 

…

loopforever: rjmp loopforever    ; Infinite loop 

RESET in ATmega2560 (1/2)

• Power-on Reset 

 The MCU is reset when the supply voltage is below the Power-on Reset 
threshold (VPOT).

• External Reset

 The MCU is reset when a low level is present on the RESET pin for 
longer than the minimum pulse length.

• Watchdog Reset

 The MCU is reset when the Watchdog Timer period expires and the 
Watchdog is enabled.

A RESET interrupt is used to restore the microcontroller to an 
initial state.

ATmega2560 microcontroller has five sources of reset:

RESET in ATmega2560 (2/2)

• Brown-out Reset 

 The MCU is reset when the supply voltage VCC is below the Brown-out 
Reset threshold (VBOT) and the Brown-out Detector is enabled.

• JTAG AVR Reset 

 The MCU is reset as long as there is a logic one in the Reset Register, one 
of the scan chains of the JTAG system. 

For each RESET, there is a flag (bit) in MCU Control Register 
MCUCSR.

• These bits are used to determine the source of the RESET interrupt. 

RESET Logic in ATmega2560 

MCU Status Register in ATmega2560 (1/3)

Bit 4 – JTRF: JTAG Reset Flag

This bit is set if a reset is being caused by a logic one in the JTAG
Reset Register selected by the JTAG instruction AVR_RESET. This bit
is reset by a Power-on Reset, or by writing a logic zero to the flag.

Bit 3 – WDRF: Watchdog Reset Flag

This bit is set if a Watchdog Reset occurs. The bit is reset by a Power-
on Reset, or by writing a logic zero to the flag.

MCU Status Register in ATmega2560 (2/3)

Bit 2 – BORF: Brown-out Reset Flag

This bit is set if a Brown-out Reset occurs. The bit is reset by a
Power-on Reset, or by writing a logic zero to the flag.

Bit 1 – EXTRF: External Reset Flag

This bit is set if an External Reset occurs. The bit is reset by a
Power-on Reset, or by writing a logic zero to the flag.

MCU Status Register in ATmega2560 (3/3)

Bit 0 – PORF: Power-on Reset Flag This bit is set if a Power-on
Reset occurs. The bit is reset only by writing a logic zero to the flag.
To make use of the Reset Flags to identify a reset condition, the user
should read and then

Watchdog Timer in ATmega2560 
• Used to detect software crash.

• Can be enabled or disabled by properly updating WDCE bit and 
WDE bit in Watchdog Timer Control Register WDTCR.

• 10 different periods determined by WDP3, WDP2, WDP1 and WDP0 
bits in WDTCR.

• If enabled, it generates a Watchdog Reset interrupt when its period 
expires.

• So program needs to reset it before its period expires by executing 
instruction WDR.

• When its period expires, Watchdog Reset Flag WDRF in MCU 
Control Register MCUCSR is set.

 This flag is used to determine if the watchdog timer has generated a RESET 
interrupt.

Watchdog Timer Logic  

50

Watchdog Timer Period  

Watchdog Timer Control Register (1/4)  

Bit 7 - WDIF: Watchdog Interrupt Flag

This bit is set when a time-out occurs in the Watchdog Timer and
the Watchdog Timer is configured for interrupt. WDIF is cleared by
hardware when executing the corresponding interrupt handling
vector. Alternatively, WDIF is cleared by writing a logic zero to the
flag. When the I-bit in SREG and WDIE are set, the Watchdog
Time-out Interrupt is executed.

Bit 6 - WDIE: Watchdog Interrupt Enable

When this bit is written to one and the I-bit in the Status Register
is set, the Watchdog Interrupt is enabled. If WDE is cleared in
combination with this setting, the Watchdog Timer is in Interrupt
Mode, and the corresponding interrupt is executed if time-out in
the Watchdog Timer occurs. If WDE is set, the Watchdog Timer
is in Interrupt and System Reset Mode. The first time-out in the
Watchdog Timer will set WDIF. Executing the corresponding
interrupt vector will clear WDIE and WDIF automatically by
hardware (the Watchdog goes to System Reset Mode).

Watchdog Timer Control Register (2/4)  

Watchdog Timer Control Register (3/4)  

Bit 4 - WDCE: Watchdog Change Enable

This bit is used in timed sequences for changing WDE and prescaler
bits. To clear the WDE bit, and/or change the prescaler bits, WDCE
must be set. Once written to one, hardware will clear WDCE after
four clock cycles.

Bit 3 - WDE: Watchdog System Reset Enable

WDE is overridden by WDRF in MCUSR. This means that WDE is
always set when WDRF is set. To clear WDE, WDRF must be
cleared first.

Bit 5, 2:0 - WDP3:0: Watchdog Timer Prescaler 3, 2, 1 and 0

The WDP3:0 bits determine the Watchdog Timer prescaling
when the Watchdog Timer is running.

Watchdog Timer Control Register (4/4)  

Timer Interrupt (1/2)

• Used to schedule (real-time) tasks (threads)

 Round-Robin scheduling

All tasks take turn to execute for some fixed period. 

 Real-time scheduling

 Some tasks must be started at a particular time and finished by a 
deadline.

 Some tasks must be periodically executed.

• Used to implement a clock

 How much time has passed since the system started?

Timer interrupt has many applications:

Timer Interrupt (2/2)

• Used to synchronize tasks.

 Task A can be started only if a certain amount of time has passed 
since the completion of task B.

• Can be coupled with a wave-form generator to support  
Pulse-Width Modulation (PWM).

 Details to be covered in the lecture about Analog Input and Output.

Timer0 in ATmega2560

• Glitch-free, Phase Correct Pulse Width Modulators (PWM).

• 10-bit Clock Prescaler.

• The counter can count up or count down depending on the 
specific operation mode.

• One Overflow Interrupt source (TOV0)  and two Compare Match 
Interrupt sources (OC0A and OC0B).

 The timer can generate a Timer0 Overflow Interrupt TOV0 when the 
counter TCNT0 overflows.

 The timer can generate two Timer/Counter0 Output Match Interrupts:  
Timer/Counter Compare Match A interrupt and Timer/Counter Compare 
Match B interrupt when TCNT0 contains the same value as in Output 
Compare Registers (OCR0A and OCR0B) 

 TOV0 and two Timer/Counter0 Output Match Interrupts can be 
individually enabled/disabled.

The 8-bit timer has the following features:

Timer0 in ATmega2560–Block Diagram

Prescaler for Timer0

Timer0 Operation Modes (1/2)

• Timer0 can operate in different operation modes.

• An operation mode specifies the behaviour of the timer and 
the Output Compare pins. 

• The behaviour of Output Compare pins and PWM modes will be 
covered in the lecture about Analog Input and Output.

• A mode is defined by the combination of the Waveform 
Generation mode (WGM2:0 in TCCR0A and TCCR0B 
registers) and Compare Output mode (COM0x1:0 in TCCR0A 
and TCCR0B registers) (x=A or B) bits. The COM0x1:0 bits 
control whether the output should be set, cleared, or toggled at 
a Compare Match.

Timer0 Operation Modes (2/2)

• The simplest operation mode is the normal mode 
(WGM02:0 = 000 in TCCR0A and TCCR0B registers), 
which is typically used to generate an interrupt, either a 
Timer Overflow Interrupt, or a Compare Match Interrupt.

• In the normal mode, the counting direction is always up 
(incrementing). The counter simply overruns when it passes 
its maximum 8-bit value 0xFF and then restarts from 0x00. 

 We focus on the normal mode now, and will learn the other modes in 
the lecture about Analog Input and Output. 

Timer0 Registers (1/8)

I/O registers for Timer0:

• Timer/Counter Register TCNT0.

 Contains the current timer/counter value.

 The value is increased or decreased by one every clock cycle based on 
the specific operation mode.

61

62



32

Timer0 Registers (2/8)

• Output Compare Registers OCR0A and OCR0B.

 Each contains an 8-bit value set by the application program. 

 Timer0 continuously compares the value of OCR0A (OCR0B) with 
the counter value of TCNT0.

When the two values of TCNT0 and OCR0A (OCR0B) are equal, the 
timer may generate a Timer/Counter Compare Match A interrupt
(Timer/Counter Compare Match B interrupt).  

Timer0 Registers (3/8)

• Timer/Counter Control Registers TCCR0A and TCCR0B.

 Contains control bits. 

 Normal mode when WGM02:0=000.

 By default, WGM02:0=000.

Timer0 Registers (4/8)

• Bits CS02:0 select the clock for Timer0.

• No prescaling when CS02=001.

Timer0 Registers (5/8)

• Timer/Counter Interrupt Mask Register TIMSK0.

 Contains interrupt enable/disable bits.

 Bit 2 – OCIE0B: Timer/Counter Output Compare Match B Interrupt 
Enable bit.

 When the OCIE0B bit is written to one, and the I-bit in the Status Register 
SREG is set, the Timer/Counter Compare Match B interrupt is enabled. The 
corresponding interrupt is executed when a Compare Match in 
Timer/Counter occurs. 

Timer0 Registers (6/8)

• Timer/Counter Interrupt Mask Register TIMSK0.

 Bit 1 – OCIE0A: Timer/Counter0 Output Compare Match A Interrupt 
Enable bit.

 When the OCIE0A bit is written to one, and the I-bit in the Status Register 
is set, the Timer/Counter0 Compare Match A interrupt is enabled. The 
corresponding interrupt is executed if a Compare Match in Timer/Counter0 
occurs.

 Bit 0 – TOIE0: Timer/Counter0 Overflow Interrupt Enable bit.

When the TOIE0 bit is written to one, and the I-bit in the Status Register is 
set, the Timer/Counter0 Overflow interrupt is enabled. The corresponding 
interrupt is executed if an overflow in Timer/Counter0 occurs.

Timer0 Registers (7/8)

• Timer/Counter Interrupt Flag Register TIFR0.

 Contains interrupt flags.

 Bit 2 – OCF0B: Timer/Counter 0 Output Compare B Match Flag.

 The OCF0B bit is set when a Compare Match occurs between the 
Timer/Counter and the data in OCR0B – Output Compare Register0 B. 
OCF0B is cleared by hardware when executing the corresponding interrupt 
handler. Alternatively, OCF0B is cleared by writing a logic zero to the flag. 
When the I-bit in SREG, OCIE0B (Timer/Counter Compare B Match 
Interrupt Enable), and OCF0B are set, the Timer/Counter Compare Match 
Interrupt is executed.

Timer0 Registers (8/8)

• Timer/Counter Interrupt Flag Register TIFR0.

 Bit 1 – OCF0A: Timer/Counter 0 Output Compare A Match Flag.

 The OCF0A bit is set when a Compare Match occurs between the 
Timer/Counter0 and the data in OCR0A – Output Compare Register0. 
OCF0A is cleared by hardware when executing the corresponding interrupt 
handler. Alternatively, OCF0A is cleared by writing a logic zero to the flag. 
When the I-bit in SREG, OCIE0A (Timer/Counter0 Compare Match 
Interrupt Enable), and OCF0A are set, the Timer/Counter0 Compare Match 
Interrupt is executed.

An Interrupt Handler Example (1/9)

• We design a program using AVR assembly language with the 
following behavior:

 Initially, 8 LEDs are set on.

 Every second, toggle the 8 LEDs, that is, if an LED is on, turn 
it off, and if it is off, turn it on.  

• Key question: How do we know if one second has passed?

 Answer: use Timer0 Overflow Interrupt.

 Need to know the interrupt period of Timer0 Overflow 
Interrupt.  

Time
First 
interrupt

Second 
interrupt

Third 
interrupt

… Interrupt
period

An Interrupt Handler Example (2/9)

• Assume the ATmega2560 microcontroller runs at a clock 
frequency of 16M Hz, and  the Timer0 prescalar value is set 
to 8.

 The interrupt period  = 256*8/16 = 128 us, i.e., every 128 us, 
Timer0 generates a Timer0 Overflow Interrupt.

 How many Timer0 Overflow Interrupts are generated each 
second? 

 1,000,000/128 =7812.5 interrupts  7812 interrupts 

• We use PortC to drive 8 LEDs.

An Interrupt Handler Example (3/9)

.include "m2560def.inc"

.equ PATTERN = 0b11110000  ; define a pattern for 8 LEDs

.def  temp = r16

.def  leds = r17   ; r17 stores a LED pattern

; The macro clears a word (2 bytes) in the data memory

; The parameter @0 is the memory address for that word

.macro clear

ldi YL, low(@0)     ; load the memory address to Y pointer

ldi YH, high(@0)

clr temp            ; set temp to 0 

st Y+, temp       ; clear the two bytes at @0 in SRAM

st Y, temp

.endmacro

An Interrupt Handler Example (4/9)

.dseg

SecondCounter: .byte 2     ; two-byte counter for counting seconds.

TempCounter: .byte 2       ; temporary counter used to determine if one second has passed

.cseg

.org 0x0000

jmp RESET

jmp DEFAULT     ; no handling for IRQ0.

jmp DEFAULT     ; no handling for IRQ1.

.org OVF0addr      ; OVF0addr is the address of Timer0 Overflow Interrupt Vector

jmp Timer0OVF   ; jump to the interrupt handler for Timer0 overflow.

…

jmp DEFAULT     ; default service for all other interrupts.

DEFAULT: reti     ; no interrupt handling 

An Interrupt Handler Example (5/9)

RESET: ldi temp, high(RAMEND)  ; initialize the stack pointer SP

out SPH, temp

ldi temp, low(RAMEND)

out SPL, temp

ser temp       ; set Port C as output

out DDRC, temp

rjmp main    ; jump to main program

An Interrupt Handler Example (6/9)

Timer0OVF:     ; interrupt subroutine to Timer0

in temp, SREG

push temp      ; prologue starts

push YH        ; save all conflicting registers in the prologue

push YL

push r25

push r24        ; prologue ends

; Load the value of the temporary counter

lds r24, TempCounter

lds r25, TempCounter+1

adiw r25:r24, 1      ; increase the temporary counter by one

An Interrupt Handler Example (7/9)

cpi r24, low(7812)         ; check if (r25:r24) = 7812

ldi temp, high(7812)      ; 7812 = 106/128

cpc r25, temp

brne NotSecond

com leds                  ; one second has passed, and toggle LEDs now 

out PORTC, leds

clear TempCounter      ; reset the temporary counter

; Load the value of the second counter

lds r24, SecondCounter

lds r25, SecondCounter+1

adiw r25:r24, 1            ; increase the second counter by one

An Interrupt Handler Example (8/9)
sts SecondCounter, r24

sts SecondCounter+1, r25

rjmp EndIF

NotSecond:         ; store the new value of the temporary counter

sts TempCounter, r24

sts TempCounter+1, r25

EndIF:   pop r24        ; epilogue starts

pop r25        ; restore all conflicting registers from the stack

pop YL

pop YH

pop temp

out SREG, temp

reti               ; return from the interrupt

An Interrupt Handler Example (9/9)

main:  ldi leds, 0xFF          ; main program starts here

out PORTC, leds     ; set all LEDs on at the beginning

ldi leds, PATTERN 

clear TempCounter         ; initialize the temporary counter to 0

clear SecondCounter      ; initialize the second counter to 0

ldi temp, 0b00000000

out TCCR0A, temp

ldi temp, 0b00000010

out TCCR0B, temp          ; set prescalar value to 8

ldi temp, 1<<TOIE0        ; TOIE0 is the bit number of TOIE0 which is 0   

sts TIMSK0, temp           ; enable Timer0 Overflow Interrupt

sei                                    ; enable global interrupt

loop: rjmp loop               ; loop forever

• The external interrupts are triggered by the INT7:0 pins or 
any of the PCINT23:0 pins. 
 We use the INT7:0 pins only for external interrupts.

• An external interrupt can be used to generate a software 
interrupt. 

• To enable an external interrupt INTx, the following two bits 
must be set:

 The I bit in SREG, and
 The INTx bit in the EIMSK register.

• Each external interrupt can be triggered by a falling or 
rising edge or a low level. 

 External Interrupt Control Registers EICRA (for 
INT3:0) and EICRB (for INT7:4) specify how 
external interrupts are triggered.

External Interrupts 

• EICRA – External Interrupt Control Register A
 EICRA contains bits for interrupt sense control (INT3:0).

Registers for External Interrupts (1/4)

• EICRB – External Interrupt Control Register B
 EICRB contains bits for interrupt sense control (INT7:4).

Registers for External Interrupts (2/4)

• EIMSK – External Interrupt Mask Register
 Bits 7:0 – INT7:0: External Interrupt Request 7 - 0 Enable 

bits.
 When an INT7:0 bit is written to one and the I-bit in the 

Status Register (SREG) is set to one, the corresponding 
external pin interrupt is enabled. 

Registers for External Interrupts (3/4)

81

82



42

• EIFR – External Interrupt Flag Register
 INTF7:0: External Interrupt Flags 7 – 0. 
 When an edge or logic change on the INT7:0 pin triggers an 

interrupt request, INTF7:0 is set to one. If the I-bit in SREG 
and the corresponding interrupt enable bit, INT7:0 in 
EIMSK, are set to one, the AVR microcontroller will jump to 
the interrupt vector. The flag is cleared when the interrupt 
routine is executed. These flags are always cleared when 
INT7:0 are configured as level interrupt.

Registers for External Interrupts (4/4)

An Example of External Interrupts (1/6)

Write an AVR assembly program to control 8 LEDs using INT0 
(External Interrupt 0) and INT1 (External Interrupt 1) as follows:

• If PB0 is pushed, the first 4 LEDs are on and the second 4 
LEDs are off

• If PB1 is pushed, the first 4 LEDs are off and the second 4 
LEDs are on

Board setting: connect PB0 to PD0 and PB1 to PD1 (Port D) 

83

84



43

An Example of External Interrupts (2/6)

.include "m2560def.inc"
.def temp =r16
.equ HIGH_LEDS = 0b11110000
.equ LOW_LEDS = 0b00001111
.cseg
.org 0x0
jmp RESET ; interrupt vector for RESET
.org INT0addr ; INT0addr is the address of EXT_INT0

; (External Interrupt 0)
jmp EXT_INT0 ; interrupt vector for External Interrupt 0
.org INT1addr ; INT1addr is the address of EXT_INT1

; (External Interrupt 1)
jmp EXT_INT1 ; interrupt vector for External Interrupt 1

An Example of External Interrupts (3/6)

RESET:
ldi temp, low(RAMEND) ; initialize stack pointer to point the high

; end of SRAM
out SPL, temp
ldi temp, high(RAMEND)
out SPH, temp
ser temp ; temp=0b11111111
out DDRC, temp ; Port C is set to all outputs
clr temp ; temp=0b00000000
out PORTC, temp
out DDRD, temp ; Port D is set to all inputs
out PORTD, temp

An Example of External Interrupts (4/6)

ldi temp, (2 << ISC10) | (2 << ISC00)
; The built-in constants ISC10=2 and ISC00=0 are their bit numbers in
; EICRA register
sts EICRA, temp ; temp=0b00001010, so both interrupts are

; configured as falling edge triggered interrupts
in temp, EIMSK
ori temp, (1<<INT0) | (1<<INT1) ; INT0=0 & INT1=1
out EIMSK, temp ; Enable External Interrupts 0 and 1
sei ; Enable the global interrupt
jmp main

An Example of External Interrupts (5/6)

EXT_INT0: ; Interrupt handler for External Interrupt 0
push temp
in temp, SREG
push temp
ldi temp, HIGH_LEDS
out PORTC, temp
pop temp
out SREG, temp
pop temp
reti

An Example of External Interrupts (6/6)

EXT_INT1: ; Interrupt handler for External Interrupt 1
push temp
in temp, SREG
push temp
ldi temp, LOW_LEDS
out PORTC, temp
pop temp
out SREG, temp
pop temp
reti

main: ; main does nothing but increments a counter
clr temp
loop: inc temp
rjmp loop ; An infinite loop must be at the end of the interrupt

; handler for RESET

Non-Nested Interrupts

• In AVR microcontrollers, by default, an interrupt handler 
(service routine) cannot be interrupted by another interrupt. 

 The global interrupt flag I is cleared when an interrupt occurs, 
preventing an interrupt service routine from being interrupted by 
another interrupt. 

Main 
program

Interrupt 
handler

Nested Interrupts

Main 
program

Interrupt 
handler 1

Interrupt 
handler 2

Interrupt 
handler 3

• In AVR microcontrollers, the interrupt handler of an interrupt 
can be interrupted by another interrupt if the interrupt 
handler sets the global interrupt flag I in SREG  using the 
instruction sei. 

Reading  Material

a. Overview

b. System Control and Reset. 

c. Watchdog Timer.

d. Interrupts. 

e. External Interrupts. 

f. 8-bit Time/Counter0

1. Chapter 8. Microcontrollers and Microcomputers.

2.   Read the following sections in ATmega2560 Data 
Sheet:

